
Introduction

The Sharp GP2D12 infrared ranger is able to continuously measure the distance to an object. The usable
range is 10 cm to 80 cm. The device generates an analog voltage that is a function of range, and the
output voltage can be measured by an analog-to-digital (ADC) input line on a BasicX system.

Hardware connections

Figure 1 (below) illustrates the hardware interface between the detector and BasicX system:

Figure 1

A voltage between 4.5 V and 5.5 V is needed to power the detector. The current drain is approximately 33
mA to 50 mA. The device starts operating as soon as power is applied.

The analog voltage output is connected to one of the ADC input pins on the BasicX system. The 470 Ω
resistor (shown above) is optional.

How it works

The device emits a pulsed infrared beam at a wavelength of 850 mn ± 70 nm. If an object is within range
and in line with the IR beam, reflected light forms an image on a linear CCD array in the receiver.
Triangulation is then used to determine range. Readings are updated at a rate of approximately 24 Hz.

The detector is relatively insensitive to ambient lighting, as well as reflectivity of the object being detected.
It is possible to detect relatively dark objects in full sunlight.

Figure 2

The output voltage is a nonlinear function of the distance from the object to the receiver. In figure 2, the
curve of voltage vs. distance was taken from Sharp documentation.

Note that objects less than 10 cm away can look like objects at longer distances. This ambiguity needs to
be taken into account if objects are allowed to be closer than the 10 cm threshold.

We need to find the inverse of the function shown above in order to determine distance as a function of
voltage. The curve in figure 2 is inverted and approximated in software by a sequence of 5 line segments
(figure 3, below). Linear interpolation is done between each of the 6 endpoints.

       Figure 3

Software interface

To read the sensor, procedure GetADC is called in order to read the analog voltage generated by the
device. GetADC returns nondimensional voltage, which needs to be multiplied by 5 to convert to units of
volts. Function VoltageToRange then uses interpolation to convert voltage to distance, where distance is
in units of meters.

Source code is provided in a separate file called SharpGP2D12.bas. A listing of the program is shown
below.

The example program transmits range data in ASCII format through the BasicX Com1 serial port at 19200
baud, 8 data bits, 1 stop bit, no parity.

'-------------------------------------------------------------------------------
Option Explicit

' This program reads a Sharp GP2D12 infrared object detector, and
' continuously displays range at a rate of about 3 times per second.
'-------------------------------------------------------------------------------
Public Sub Main()

    Dim Range As Single
    Dim Success As Boolean

    Debug.Print
    Debug.Print "Sharp GP2D12 IR object detector"
    Debug.Print

    Do
        Call GetRange(Range, Success)

        Debug.Print "Range = ";

        If (Success) Then
            Debug.Print CStr( CInt(Range * 1000.0) ); " mm"
        Else
            ' Out of range.
            Debug.Print " ***"
        End If

        Delay 0.3
    Loop

End Sub
'-------------------------------------------------------------------------------
Public Sub GetRange( _
    ByRef Range As Single, _
    ByRef Success As Boolean)

    Dim Voltage As Single
    Const MinVolt As Single = 0.399
    Const MaxVolt As Single = 2.60

    ' This pin number is for a BX-24. The pin number may need to be changed
    ' depending on the BasicX system being used.
    Const InputPin As Byte = 16

    Call GetADC(InputPin, Voltage)

    ' Convert to voltage.
    Voltage = Voltage * 5.0

    Range = VoltageToRange(Voltage)

    ' Check for legal voltage.
    If (Voltage >= MinVolt) And (Voltage <= MaxVolt) Then
        Success = True
    Else
        Success = False
    End If

End Sub
'-------------------------------------------------------------------------------

'-------------------------------------------------------------------------------
Private Function VoltageToRange( _
    ByVal Voltage As Single) As Single

' Returns distance in units of meters. The nonlinear curve of distance
' vs. voltage is approximated by a sequence of 5 line segments. Linear
' interpolation is done between endpoints.

    ' Lower half of voltage range.
    If (Voltage < 0.8181650) Then

        If (Voltage < 0.5860420) Then

            ' Segment 1.
            VoltageToRange = -1.431295831 * Voltage + 1.371201638
        Else
            ' Segment 2.
            VoltageToRange = -0.751946517 * Voltage + 0.973074437
        End If

    ' Upper half of voltage range.
    Else
        If (Voltage < 1.1492463) Then

            ' Segment 3.
            VoltageToRange = -0.349412673 * Voltage + 0.643735317

        ElseIf (Voltage < 1.6363301) Then

            ' Segment 4.
            VoltageToRange = -0.177086187 * Voltage + 0.445689738
        Else
            ' Segment 5.
            VoltageToRange = -0.076191604 * Voltage + 0.280592896
        End If
    End If

End Function
'-------------------------------------------------------------------------------

